

Keysight Technologies ENA Series Network Analyzer – VBA Programming (UserMenu)

Procedure overview


1. Connect DUT to ENA
2. Launch VBA editor, and code VBA program
3. Run VBA Macro
4. Add bandwidth search module
5. Apply VBA macro to UserMenu buttons

In this demo...

- Code simple VBA macro for bandpass filter measurement
- Apply VBA macro to UserMenu buttons

Required Instrument and fixture


ENA series network analyzer (E5071C or E5061B)



Band pass filter (BPF)

In this demo, we will use BPF (Center frequency = 1.09 GHz, BW = 15 MHz) but you can use another filter. Prepare appropriate cable and adapter to connect between ENA and DUT.

1. Connect DUT to E5061B




2. Launch VBA editor, and code VBA program

- a. Press **[Macro setup]** hard key then press **VBA Editor** soft key VBA editor
- b. Create module and code procedure
Click **Insert** in the menu bar then click **Module**
Code as shown below in **Module1**

```
Sub main()
    Call Setup
End Sub


Sub Setup()
    SCPI.SYSTEM.PRESet
    SCPI.SENSE.FREQUENCY.CENTer = "1.09E9"
    SCPI.SENSE.FREQUENCY.SPAN = "200E6"
    SCPI.CALCulate.PARAmeter.Count = 2
    SCPI.DISPlay.WINDow.Split = "D12"
    SCPI.CALCulate.PARAmeter(2).DEfine = "S21"
    SCPI.SENSE.BANDwidth.RESolution = 1000
    MsgBox "Setup done"
    SCPI.DISPlay.WINDow.TRACe(1).Y.SCALE.AUTO
    SCPI.DISPlay.WINDow.TRACe(2).Y.SCALE.AUTO
End Sub
```

- c. Save VBA program and exit VBA editor
Click  icon of the VBA editor. The Save As dialog box appears. Specify the file name and location and click **Save**.
Click  icon of the VBA editor.



3. Run VBA macro

- a. Press **[Macro Run]** Hard key.
The ENA calls main() procedure, and this VBA program will set up the measurement parameters as shown below.



4. Add Bandwidth Search Module


- Open VBA Editor
- Modify **Sub()** procedure of **Module1** as below code

```
Sub Main()
    Call Setup
    Call Bandwidth
End Sub
```

- Add below procedure on **Module1**

```
Sub Bandwidth()
    SCPI.CALCulate.PARAMeter(2).SElect
    With SCPI.CALCulate.SELECTed.MARKer
        .State = True
        .FUNCTION.TYPE = "MAX"
        .FUNCTION.EXECute
        .BWIDth.State = True
    End With
End Sub
```

- Save VBA program and run
Click  icon, then  icon of the VBA editor
Press **[Macro Run]** hard key.
This code sets the measurement parameter, then make bandwidth search function as below.




5. Apply VBA procedure to UserMenu buttons

- Open VBA Editor
- Modify **Sub()** procedure of **Module1** as below code

```
Sub Main()
    UserMenu.Item(1).enabled = True
    UserMenu.Item(2).enabled = True
    UserMenu.Item(1).Caption = "Setup"
    UserMenu.Item(2).Caption = "Bandwidth"
    UserMenu.Show
End Sub
```

- Create **UserMenu** module
Click E5061B_Objects then Double-Click UserMenu
In the object box in the code window, select UserMenu as shown below.




- Code below procedure on UserMenu module

```
Private Sub UserMenu_OnPress (byVal ID As Long)
    If ID = 1 Then Module1.Setup
    If ID = 2 Then Module1.Bandwidth
    UserMenu.Show
End Sub
```

- Save program and run
Click  icon, then  icon of the VBA Editor
Press **[Macro Run]** hard key.

6. Use UserMenu Buttons

- Press "Setup" button to run Setup procedure
- Press "Bandwidth" button to run Bandwidth_Search procedure



Tips: Command finder

Using command finder chapter of the help file, you can easily find appropriate command for each button. To open help, press **[Help]** hard key, then click **Programming > Command Reference > Command Finder**.

